


Outlook for Mac Exchange Setup

4.0 Administrator's Guide

William Smith
bill@officeformachelp.com

© Copyright 2008-2010 William Smith

 Except where otherwise noted, this work is licensed under
<http://creativecommons.org/licenses/by/3.0/>

Introduction

Connecting Microsoft Outlook for Mac to an Exchange Server is not always easy for the typical end-user. If he's lucky then the network has been properly configured to help him discover his Exchange Server and he can take advantage of Outlook's automatic setup feature. But that's not always the case.

DNS must be properly configured to enable Microsoft Outlook 2011 to discover its user's Exchange Server and LDAP server or it must be configured to support AutoDiscover with Exchange Server 2007 or higher. Not all environments are properly configured to do this.

Even with AutoDiscover, however, user preferences cannot be configured. Common rules, signatures, attachment settings, etc., must be set individually.

Outlook for Mac Exchange Setup 4.0 is a new set of scripts based on my Entourage Exchange Setup scripts that assist a Macintosh administrator in the pre-configuration of Outlook for Mac settings so that users will have common default preferences and will not be exposed to the details of having to set up their own Exchange accounts. Every company's network is different so this script was written to provide enough ideas for a novice scripter to be able to adapt it for use in his network in just a few minutes.

The entire user experience can be made as simple as logging in to a workstation for the first time, launching Outlook for Mac and working immediately. Just a few seconds.

How it works

Using the *Outlook for Mac Exchange Setup 4.0* scripts the Macintosh administrator will configure a new Office 2011 identity without any accounts. He can configure all preferences in that identity including preferences, schedules, signatures, rules, custom views, etc.

This pre-configured identity along with all necessary preferences will become a *template* for all future users of the workstation. The template identity will be deployed to the *User Template* folder of all workstations.

What is the User Template?

If you haven't been using the Mac OS X User Template in your arsenal of administrative tools then you're really missing something!

The User Template folder is located in /System/Library. Inside this folder are all the default files and folders for a new user account. When a user logs in to a Mac workstation for the first time all the items from the User Template folder are copied to his new home folder. Those folders include Desktop, Documents, Library, Movies, Music, Pictures, Public and Sites.

A Mac administrator can take advantage of the User Template folder by populating it with documents, settings, application preferences and more. All new home folders can be customized with default data specific to a group or entire organization.

When a user logs in to his Mac workstation for the first time the Office identity will be copied from the User Template to his new home folder.

When the user launches Outlook for Mac for the first time an Outlook Schedule will trigger the *Outlook for Mac Exchange Setup 4.0* script. The script will configure the user's Exchange settings and then remove the schedule.

From the user's point-of-view he simply launches Outlook for Mac and begins working with preferences and settings the administrator has pre-defined.

Prepare the Script

This section will guide you through configuring the *Outlook for Mac Exchange Setup 4.0* script.

Server, Schedule and Custom Message Properties

Locate the *Outlook for Mac Exchange Setup 4.0* file and double-click it to open it in the *AppleScript Editor.app* application found in the */Applications/Utilities* folder.

The beginning of the script contains comments for the reader between a set of **(*** and ***)** markers. Everything between these markers is ignored when the script is executed.

Throughout the script are additional comments that begin with “**--**” (double-hyphens or multiple hyphens. These are ignored when the script is executed.

The only section that the Mac administrator will need to customize is between:

```
-----  
-- Begin network and server properties  
-----
```

and

```
-----  
-- End server properties  
-----
```

The Mac administrator can customize 19 *properties* in this section. Each property is a setting that the script will use later when executed. The properties are named so that they are easily identifiable and are explained by the comments below them.

Every property corresponds to an Exchange account setting or identity setting. The **Accounts** pane is found under **Outlook** menu → **Preferences...** (See Figures 1 -2) The **Schedules** pane is found under **Outlook** menu → **Preferences...** (See Figure 3)




Figure 1

A	<pre>property useKerberos : false -- Set this property to true only if Macs in your environment -- are bound to Active Directory. -- Kerberos support works only if your network is properly -- configured.</pre>
B	<pre>property kerberosRealm : "example.com" -- For most Kerberos environments the Kerberos realm -- will be the same as the E-mail domain name. -- This property is used only if the userKerberos property above -- is true.</pre>




Figure 2

C	<pre>property domainName : "example.com" -- example: "domain.com"</pre>
----------	---




Figure 3

D	<pre>property ExchangeServer : "exchange01.mydomain.com" -- example: "ExchangeServer.domain.com"</pre>
D	<pre>property ExtendedExchangeServerAddress : false -- Are you connecting to an Exchange Server 2007 server? -- If so, you may need to use the extended address: -- "ExchangeServer.domain.com/exchange/user@example.com". -- Set this to "true" if you need the extended address.</pre>
E	<pre>property ExchangeServerRequiresSSL : true -- If Exchange users will connect to a backend server -- within the same network then SSL is probably not required. -- OWA connections will probably require SSL and -- this setting should be set to true.</pre>
F	<pre>property ExchangeServerSSLPort : 443 -- If ExchangeServerSSL is false set the port to 80. -- If ExchangeServerSSL is true set the port to 443. -- Use a different port number only if specified -- by your administrator.</pre>

G	<pre>property DirectoryServer : "gc.example.com" -- example: "GlobalCatalogServer.domain.com"</pre>
H	<pre>property DirectoryServerRequiresAuthentication : true -- This will almost always be true. The LDAP server -- in a Windows network will be a Global Catalog server, -- which is separate from the Exchange Server. -- OWA connections can not use the OWA address -- with this setting.</pre>
I	<pre>property DirectoryServerRequiresSSL : false -- If Exchange users will connect to a backend server -- within the same network then SSL is probably -- not required. OWA connections can not use -- the OWA address with this setting.</pre>
J	<pre>property DirectoryServerSSLPort : 3268 -- If DirectoryServerRequiresSSL is false set the port to 3268. -- If DirectoryServerRequiresSSL is true set the port to 3269. -- Use a different port number only if specified -- by your administrator. -- OWA connections cannot use the OWA address with this setting.</pre>
K	<pre>property DirectoryServerMaximumResults : 1000 -- When searching the Global Address list, this number determines -- the maximum number of entries to display.</pre>
L	<pre>property DirectoryServerSearchBase : "" -- example: "cn=users,dc=domain,dc=com" -- Search base will be optional in many environments and its -- format will vary greatly. Experiment first connecting without -- entering the search base information.</pre>
M	<pre>property getEmailAddressUsingDSCL : true -- If the Macs are connected to a directory service such as -- Active Directory, then they can probably use dscl to return -- the current user's E-mail address instead of trying to parse it -- from the display name. -- Using dscl is preferred. Otherwise, set this to false -- and set the next property to the appropriate number.</pre>
N	<pre>property emailFormat : 1 -- When dscl is unavailable to determine a user's E-mail address, -- it can be parsed using the display name of the user's login. -- Set the next two properties to the examples -- that correspond to your organization's setup. -- 1: E-mail format is first.last@domain.com -- 2: E-mail format is first@domain.com -- 3: E-mail format is flast@domain.com -- (first name initial plus last name) -- 4: E-mail format is shortName@domain.com property displayName : 1 -- 1: Display name displays as "Last, First" -- 2: Display name displays as "First Last"</pre>




Figure 4

0	<pre>property LDAPSearchBase : "" -- example: "cn=users,dc=domain,dc=com" -- Search base will be optional in many environments and its -- format will vary greatly. Experiment first connecting without -- entering the search base information.</pre>
---	---

By default, the script requires no user interaction. However, if users need to specify a different E-mail server from the standard or a uses a custom E-mail address, then these properties can be set to true to prompt the user to verify the correct settings.

	<pre>property verifyEmailAddress : false -- If set to "true", a dialog will ask the user to confirm -- his E-mail address.</pre>
	<pre>property verifyServerAddress : false -- If set to "true", a dialog will ask the user to confirm -- his Exchange server address.</pre>

Disable the **Send & Receive All** schedule if users only have Exchange accounts. This feature is necessary for POP and IMAP accounts only.

	<pre>property scheduled : false -- Exchange accounts don't require that -- the "Send & Receive All" schedule be enabled. -- Change this setting to true if the user -- will also be connecting to POP or IMAP accounts.</pre>
--	---

Should the script fail the Mac administrator can insert a custom error message directing the user to contact the company Help Desk or take some other action.

	<pre>property errorMessage : "Setup of your account failed. Please contact your workstation administrator for assistance." -- Customize this error message for your users -- if their account setup fails</pre>
--	---

The Rest of the Script

The rest of the script requires no modifications by the Mac administrator unless he chooses to augment its function. If left as-is then it will proceed to:

1. Determine the user name and display name of the current user.
2. Determine the current user's E-mail address by using dscl or parsing it from login information.
3. Ask the user to verify his E-mail address and Short Name if those options are enabled.
4. Configure the Exchange account.
5. Enable Kerberos if that option is enabled and set the principal name as short@kerberosRealm.
6. Set the **Send & Receive All** schedule to disabled unless specified in the above properties that it should be enabled. (*See Figure 4*)
7. Delete the "Exchange Setup" schedule that triggers the script at launch.
8. Alert the user if an error occurs.

Saving the Script

Save the script as **File Format: Script** using *Outlook for Mac Exchange Setup 4.0* or another more appropriate name. Users will not see this script. (See *Figure 5*)




Figure 5

Create the Outlook template files

This section will guide you through the process of creating a “template” **Microsoft User Data (MUD)** folder and the accompanying preference files that you will deploy to the User Template folder on your Mac workstations.

Install Microsoft Office 2011

These scripts are not responsible for deploying Outlook to the Mac workstations. That can be accomplished with a variety of tools such as Apple Remote Desktop (ARD), Casper (JAMF Software) or InstaDMG (AFP548.com).

The Outlook template files must be prepared on a Mac workstation that has a pre-existing version of Microsoft Office 2011, preferably patched to the latest available level. Enter all necessary Product Key and User registration information. Launch Outlook for Mac to verify the software was installed correctly.

Create a Setup Account

To get a clean set of files you should create a new Mac OS X user account. Select **Apple** menu → **System Preferences...** → **Accounts** and create a new Standard user account. This can be a temporary account.

Log in to the new user account.

Configure Outlook Preferences and Settings

This is the most important step of the process because you can customize your users’ default settings in every way. This includes preferences, schedules, signatures, rules, custom views, etc.

First, let’s set some needed preferences:

Launch Outlook for Mac. Then, set Outlook to work offline by selecting **Outlook** menu → **Work Offline**. This will prevent Outlook from trying to make any network connections while the script is running.

Preferences: Select **Outlook** menu → **Preferences...** Take this time to browse through each setting on each screen, configuring those settings you'd like to set as the defaults for your users.

Message lists: The default view for each folder will propagate to all new sub-folders, however, if you want to change all existing folders the changes must be made manually on each.

You can customize columns in both the Inbox and Sent Items folder. You may want to make sure that the **Sent** column is displayed and that the **Received** column is hidden.

Rules: Outlook has three types of rules—Junk E-mail Protection, Mailing List Manager and Rules. Consider setting Junk E-mail Protection to Low or None if you know your server-side filters offer good protection. This will reduce the risk of legitimate customer or co-worker mail messages getting flagged.

Accounts: If your company uses an LDAP server then configure this setting under **Outlook** menu → **Preferences...** → **Accounts...** → Directory Service.

Signatures: Does your company have a standard signature they prefer all users to append to their mail messages? Configure this setting under **Outlook** menu → **Preferences...** → Signatures...

Custom Welcome message: Consider creating a *Welcome!* message. You can include “New User” information such as important telephone numbers, campus locations, what to do if a printer gets jammed and how to contact the Help Desk.

Once your default settings are set be sure to empty the **Deleted Items** folder and clear any traces of your presence.

Create the “Exchange Setup” Schedule

Now, you're ready to hook into the *Outlook for Mac Exchange Setup 4.0* script.

Copy the script to a location on your workstation where all users will have read access. The most appropriate place would be in the /Library folder at the root of the startup disk. Consider installing it in:

```
/Library/CompanyName/Outlook for Mac Exchange Setup 4.0
```

Set permissions on the script file to:

```
-rw-rw-r-- root admin 664
```

or just propagate the enclosing folder's permissions down to the file.

In Outlook, select **Outlook** menu → **Preferences...** → Schedules and create a new schedule. (See figure 5)




Figure 6

- Name the schedule *exactly* “Exchange Setup”. The setup script looks for this name when deleting the schedule at the end of the execution.
- Set **When** to run **At Startup**.
- Set one **Action** to **Run AppleScript**.
- Click the Script... button and select the *Outlook for Mac Exchange Setup 4.0* script.
- Make sure the schedule is **Enabled**.
- Click the **OK** button and save the schedule.

The next time Outlook is launched the Startup Schedule “Exchange Setup” will run and begin the account setup for the user. When it completes successfully the script will delete this schedule so that it doesn’t run again.

Test the Setup

Everything is nearly in place. The next step is to copy the new Microsoft User Data (MUD) folder with the Outlook identity and the preferences that were created along with it to the local User Template folder for testing.

Copy the Outlook Template Files to User Template

Locate the *Copy Outlook Files.app* application that was downloaded with the *Outlook for Mac Exchange Setup 4.0* script. This AppleScript application does one of two things:

- It copies the Outlook Template files to the local computer's User Template folder.
- It copies the Outlook Template files to a folder you specify.

For testing the setup you will select the option to copy the Outlook Template files to local computer's User Template folder.

First, make sure Outlook is quit. Then double-click the *Copy Outlook Files.app* application and click the **User Template folder** button. (See Figure 7)



Figure 7

The script just takes a few seconds to run. In that time it copies the following files:

From: ~/Documents
To: /System/Library/User Template/English.lproj/Documents
“Microsoft User Data” folder and its contents

From: ~/Library/Preferences
To: /System/Library/User Template/English.lproj/Library/Preferences
“Microsoft” folder and its contents com.apple.LaunchServices.plist all com.microsoft. files

The com.apple.LaunchServices.plist file is actually created by the script. This file sets the default mail application to Outlook so that Outlook will not request that it be made the default.

Everything should be ready.

Create a second new Mac OS X user account. Select **Apple** menu → **System Preferences...** → **Accounts** and create a new Standard user account. This can be a temporary account.

Either log out of the current account or use Fast User Switching to change to the second Mac OS X user account. The files from the User Template folder should have populated the second Mac OS X user account’s home folder. Verify that the MUD folder is in the current user’s ~/Documents folder. Verify that the Microsoft preference files are in the current user’s ~/Library/Preferences folder.

When everything is in place then launch Outlook for Mac. The user should be prompted to enter his password to connect to his Exchange account.

Verify that Outlook preferences and account settings are correct. If you find that you've missed a setting or want to change a setting log in to the first new Mac OS X user account and make the corrections in Outlook.

After you make the corrections, quit Outlook and run the *Copy Outlook Files.app* application again. Newer files will be copied over the older files in the User Template folder.

Create a new Mac OS X user account. (Do not reuse accounts. User Template files are only copied once to new accounts.) Test your settings again.

If you are satisfied that the setup is working correctly then you are ready to collect the files for packaging.

Collect the Outlook Template files for Distribution

The final step is to collect the Outlook template files to a location where you can package them for distribution.

Packaging the template files is beyond the scope of these scripts. You can use any of several methods for distributing the collected files. Below are a few options:

- Apple Remote Desktop
- Apple's PackageMaker.app application
- JAMF Software's Casper Suite
- AFP548.com's InstaDMG scripts

Collect the Outlook Template files

You will use the *Copy Outlook Files.app* application again to collect the files but this time you will specify a folder of your choice.

First, make sure you are logged in to your first new Mac OS X account and that Outlook is quit. Then double-click the *Copy Outlook Files.app* application and click the **Another folder** button. (See Figure 8)



Figure 8

Select a folder or create a new folder and click the **Choose** button. The same files that were copied to the User Template folder will be copied to your new folder.

Package these files and distribute them to the User Template folders on your Mac workstations. Be sure to include the *Outlook for Mac Exchange Setup 4.0* script with your package. Remember, you are responsible for installing Office 2011 separately.

Appendix A

The Outlook for Mac Exchange Setup 4.0 script:

(*

```
-----
Outlook Exchange Setup 4.0.1
© Copyright 2008-2010 William Smith
bill@officeformachelp.com
```

Except where otherwise noted, this work is licensed under
<http://creativecommons.org/licenses/by/3.0/>

This script may be freely modified for personal or commercial
 purposes but may not be republished for profit without prior consent.

If you find this script useful or have ideas for improving it,
 please let me know. It is only compatible with Outlook for Mac 2011.

```
-----
This script assists a user with the setup of his Exchange account
information. Below are basic instructions for using the script.
Consult the Outlook Exchange Setup 4.0 Administrator's Guide
for complete details.
```

1. Customize the "network and server properties" below with information appropriate to your network.
2. a. Launch Microsoft Outlook under a Mac OS X account that has no Outlook identity. Identities are found in
 ~/Documents/Microsoft User Data/Office 2011 Identities.
- b. Close the Outlook Setup Assistant window and select
 Outlook --> Work Offline.
- c. In the Finder, create a folder called "Outlook Setup Script"
 in /Library/<your company name>/ or any single location
 where all users have Read permissions.
- d. Copy your customized script to the "Outlook Setup Script" folder.
 It should be saved as a compiled script, not an application.
 You can rename the script if you wish.
- e. In Outlook select Tools --> Schedules... and create a New
 Schedule.

Name: Outlook Setup
When: At Startup
Action: Run AppleScript, choose the saved script
in the "Outlook Setup Script" folder.
Click the "OK" button. Your script should be set to run
at "Next Startup".

- f. Configure any preferences, schedules, signatures, etc.
This can include Default address format, Default time zone,
fonts, encoding for attachments and more.
- g. Quit Outlook.
- h. Depending on your admin tools, you can push the script to the
/Library folder of a new computer and the
"~/Documents/Microsoft User Data/Office 2011 Identities/Main
Identity" folder and to the "/System/Library/User
Template/English.lproj/Documents/Microsoft User Data/
Office 2011 Identities" folder.

New users on a machine will have the pre-configured Main Identity
folder copied to their home folders the first time they log in to
a computer.

The first time they launch Outlook, the "Exchange Setup" schedule
will run the script.

The script will disable the Schedule once it has run and will set
Outlook to work online.

This script assumes the user's full name is in the form of "Last,
First", but is easily modified if the full name is in the form of
"First Last". It works especially well if the Mac is bound to Active
Directory where the user's short name will match his login name.
Optionally, a dscl command can be used to pull the user's
EmailAddress from a directory service.

*)

```
-----  
-- Begin network and server properties  
-----
```

property useKerberos : false

-- Set this property to true only if Macs in your environment
-- are bound to Active Directory.
-- Kerberos support works only if your network is properly
-- configured.

property kerberosRealm : "example.com"

-- For most Kerberos environments the Kerberos realm
-- will be the same as the E-mail domain name.
-- This property is used only if the userKerberos property above
-- is true.

```

property domainName : "example.com"
-- example: "domain.com"

property ExchangeServer : "exchange.example.com"
-- example: "ExchangeServer.domain.com"

property ExtendedExchangeServerAddress : false
-- Are you connecting to an Exchange Server 2007 server?
-- If so, you may need to use the extended address:
-- "ExchangeServer.domain.com/exchange/user@example.com".
-- Set this to "true" if you need the extended address.

property ExchangeServerRequiresSSL : true
-- If Exchange users will connect to a backend server
-- within the same network then SSL is probably not required.
-- OWA connections will probably require SSL and
-- this setting should be set to true.

property ExchangeServerSSLPort : 443
-- If ExchangeServerSSL is false set the port to 80.
-- If ExchangeServerSSL is true set the port to 443.
-- Use a different port number only if specified
-- by your administrator.

property DirectoryServer : "gc.example.com"
-- example: "GlobalCatalogServer.domain.com"

-- OWA connections cannot use the OWA address.

property DirectoryServerRequiresAuthentication : true
-- This will almost always be true. The LDAP server
-- in a Windows network will be a Global Catalog server,
-- which is separate from the Exchange Server.
-- OWA connections can not use the OWA address
-- with this setting.

property DirectoryServerRequiresSSL : false
-- If Exchange users will connect to a backend server
-- within the same network then SSL is probably
-- not required. OWA connections can not use
-- the OWA address with this setting.

property DirectoryServerSSLPort : 3268
-- If DirectoryServerRequiresSSL is false set the port to 3268.
-- If DirectoryServerRequiresSSL is true set the port to 3269.
-- Use a different port number only if specified
-- by your administrator.

-- OWA connections cannot use the OWA address with this setting.

property DirectoryServerMaximumResults : 1000
-- When searching the Global Address list, this number determines
-- the maximum number of entries to display.

property DirectoryServerSearchBase : ""

```

```
-- example: "cn=users,dc=domain,dc=com"

-- Search base will be optional in many environments and its
-- format will vary greatly. Experiment first connecting without
-- entering the search base information.

property getEmailAddressUsingDSCL : true
-- If the Macs are connected to a directory service such as
-- Active Directory, then they can probably use dscl to return
-- the current user's E-mail address instead of trying to parse it
-- from the display name.

-- Using dscl is preferred. Otherwise, set this to false
-- and set the next property to the appropriate number.

property emailFormat : 1
-- When dscl is unavailable to determine a user's E-mail address,
-- it can be parsed using the display name of the user's login.
-- Set the next two properties to the examples
-- that correspond to your organization's setup.

-- 1: E-mail format is first.last@domain.com
-- 2: E-mail format is first@domain.com
-- 3: E-mail format is flast@domain.com
--    (first name initial plus last name)
-- 4: E-mail format is shortName@domain.com

property displayName : 1
-- 1: Display name displays as "Last, First"
-- 2: Display name displays as "First Last"

property verifyEmailAddress : false
-- If set to "true", a dialog will ask the user to confirm
-- his E-mail address.

property verifyServerAddress : false
-- If set to "true", a dialog will ask the user to confirm
-- his Exchange server address.

property scheduled : false
-- Exchange accounts don't require that
-- the "Send & Receive All" schedule be enabled.
-- Change this setting to true if the user
-- will also be connecting to POP or IMAP accounts.

property errorMessage : "Setup of your account failed. Please contact
    your workstation administrator for assistance."
-- Customize this error message for your users
-- if their account setup fails

-----
-- End server properties
-----

-----
```

```

-- Begin get user name
-----

-- User information is pulled from the account settings of the current
  user account

tell application "System Events"
    set shortName to name of current user
    set fullName to full name of current user
end tell

-----

-- End get user name
-----

-----

-- Begin email address properties
-----

if getEmailAddressUsingDSCL is true then
    set emailAddress to do shell script "dscl . -read /Users/" &
    shortName & " EMailAddress | awk 'BEGIN {FS=\"\": \"\"} {print $2}'"

else if emailFormat is 1 and displayName is 1 then

    -- first.last@domain.com and full name displays as "Last, First"

    set AppleScript's text item delimiters to ", "
    set firstName to text item 2 of fullName
    set lastName to text item 1 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to firstName & "." & lastName & "@" & domainName

else if emailFormat is 1 and displayName is 2 then

    -- first.last@domain.com and full name displays as "First Last"

    set AppleScript's text item delimiters to " "
    set firstName to text item 1 of fullName
    set lastName to text item 2 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to firstName & "." & lastName & "@" & domainName

else if emailFormat is 2 and displayName is 1 then

    -- first@domain.com and full name displays as "Last, First"

    set AppleScript's text item delimiters to ", "
    set firstName to text item 2 of fullName
    set lastName to text item 1 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to firstName & "@" & domainName

else if emailFormat is 2 and displayName is 2 then

```

```
-- first@domain.com if full name displays as "First Last"

set AppleScript's text item delimiters to " "
set firstName to text item 1 of fullName
set lastName to text item 2 of fullName
set AppleScript's text item delimiters to ""
set emailAddress to firstName & "@" & domainName

else if emailFormat is 3 and displayName is 1 then

    -- flast@domain.com and full name displays as "Last, First"

    set AppleScript's text item delimiters to ", "
    set firstName to text item 2 of fullName
    set lastName to text item 1 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to (character 1 of firstName) & lastName & "@" &
    domainName

else if emailFormat is 3 and displayName is 2 then

    -- flast@domain.com and full name displays as "First Last"

    set AppleScript's text item delimiters to " "
    set firstName to text item 1 of fullName
    set lastName to text item 2 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to (character 1 of firstName & lastName & "@" &
    domainName)

else if emailFormat is 4 and displayName is 1 then

    -- shortName@domain.com and full name displays as "Last, First"

    set AppleScript's text item delimiters to ", "
    set firstName to text item 2 of fullName
    set lastName to text item 1 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to shortName & "@" & domainName

else if emailFormat is 4 and displayName is 2 then

    -- shortName@domain.com and full name displays as "First Last"

    set AppleScript's text item delimiters to " "
    set firstName to text item 1 of fullName
    set lastName to text item 2 of fullName
    set AppleScript's text item delimiters to ""
    set emailAddress to shortName & "@" & domainName

end if

-----
-- End email address properties
-----
```

```

-----
-- Exchange Server 2007 extended address
-----

if ExtendedExchangeServerAddress is true then
    set ExchangeServer to ExchangeServer & "/exchange/" & emailAddress
end if

-----

-- End get user name
-----

-----

-- Begin account setup
-----

try

    if verifyEmailAddress is true then
        set verifyEmail to display dialog "Please verify that your email
        address is correct." default answer emailAddress with icon 2
        buttons {"Cancel", "OK"} default button {"OK"}
        set emailAddress to text returned of verifyEmail
    end if

    if verifyServerAddress is true then
        set verifyServer to display dialog "Please verify that your
        Exchange Server name is correct." default answer ExchangeServer
        with icon 2 buttons {"Cancel", "OK"} default button {"OK"}
        set ExchangeServer to text returned of verifyServer
    end if

    tell application "Microsoft Outlook"
        activate
        set newExchangeAccount to make new exchange account with
        properties -
            {name:"Mailbox - " & fullName, user name:shortName, full
            name:fullName, email address:emailAddress,
            server:ExchangeServer, use ssl:ExchangeServerRequiresSSL,
            port:ExchangeServerSSLPort, ldap server:DirectoryServer,
            ldap needs
            authentication:DirectoryServerRequiresAuthentication, ldap
            use ssl:DirectoryServerRequiresSSL, ldap max
            entries:DirectoryServerMaximumResults, ldap search
            base:DirectoryServerSearchBase}

        -- The following lines enable Kerberos support if the
        userKerberos property above is set to true.

        if useKerberos is true then
            set use kerberos authentication of newExchangeAccount to
            useKerberos
            set principal of newExchangeAccount to shortName & "@" &
            kerberosRealm
        end if
    end tell
end try

```

```
end if

-- The Me Contact record is now automatically created with the
  first account.
-- Set the first and last name of the Me Contact record.

set first name of me contact to firstName
set last name of me contact to lastName
set email addresses of me contact to
  {address:emailAddress, type:work}

-- We're done. Set the schedules back to normal and set Outlook
  to work online

set enabled of schedule "Send & Receive All" to scheduled
delete schedule "Exchange Setup"
set working offline to false

end tell

on error

  display dialog errorMessage with icon 2 buttons {"OK"} default
    button {"OK"}

end try

-----
-- End account setup
-----
```


Appendix B

The Copy Outlook Files.app script:

```
(*
-----
Outlook Exchange Setup 4.0
© Copyright 2008-2010 William Smith
bill@officeformachelp.com

Except where otherwise noted, this work is licensed under
  http://creativecommons.org/licenses/by/3.0/

This script may be freely modified for personal or commercial purposes
  but may not be republished for profit without prior consent.

If you find this script useful or have ideas for improving it, please
  let me know. It is only compatible with Outlook 2011 for Mac.

This script is used in conjunction with the OutlookExchangeSetup4.0.scpt
  script or later.

1. Configure the OutlookExchangeSetup4.0.scpt file with your Exchange
  Server settings

2. Copy the OutlookExchangeSetup4.0.scpt file to a commonly accessible
  folder on your computer such as /Library/yourCompanyName. Make sure
  all users have read permission for the script file..

3. Create a new user account.

4. Launch Microsoft Outlook for Mac and set up and configure all
  preferences, schedules, rules, etc., that all users should have by
  default.

5. Create a new schedule using Tools --> Schedules...

  Name it exactly "Exchange Setup".
  Set it to run "At Startup".
  Add Action "Run AppleScript" and select your pre-configured
  OutlookExchangeSetup4.0.scpt file.
  Click the OK button and to save the "Exchange Setup" schedule. It
  should be enabled.

6. Run this script. It does three things:
```

- a.) It will copy the current Microsoft User Data folder to the User Template folder.
 - b.) It will copy all Microsoft preferences to the User Template folder.
 - c.) It will set the default mail application to Outlook in the User Template folder.
7. To test, create a new user account on the same computer and log in. Launch Microsoft Outlook for Mac. Your new user account should run the script and then delete the "Exchange Setup" schedule.

*)

-- Where do you want to copy your Outlook files?

set folderChoice **to display dialog** "This script will copy the default set of Outlook files to your computer's User Template folder or to a folder of your choice.

Where would you like to copy the files?" with title "Copy files" with icon 1 buttons {"Cancel", "Another folder", "User Template folder"} default button {"User Template folder"}

if button returned **of** folderChoice **is** "Another folder" **then**
 copyToAnotherFolder()
end if

if button returned **of** folderChoice **is** "User Template folder" **then**
 copyToUserTemplateFolder()
end if

on copyToUserTemplateFolder()

-- Get the path to the current home folder
-- in the form of /Users/homefolder

set homeFolder **to** POSIX path **of** (path to home folder)

-- copy the Microsoft User Data folder from the current user's home folder
-- to /System/Library/User Template/English.lproj/Documents

set MUD **to** "\" & homeFolder & "Documents/Microsoft User Data" & "\"

set UTDDocuments **to** "\" & "/System/Library/User Template/English.lproj/Documents" & "\"

set copyMUD **to** "cp -fR " & MUD & " " & UTDDocuments **as string**
do shell script copyMUD **with** administrator privileges

-- copy the Microsoft preferences from the current user's home folder
-- to /System/Library/User Template/English.lproj/Library/Preferences

```

set MSPreferences to homeFolder &
    "Library/Preferences/*[Mm]icrosoft*"
set UTPreferences to "\"" & "/System/Library/User
    Template/English.lproj/Library/Preferences" & "\""
set copyMSPreferences to "cp -fR " & MSPreferences & " " &
    UTPreferences as string
do shell script copyMSPreferences with administrator privileges

do shell script "defaults write
    /System/Library/User\\Template/English.lproj/Library/Preferences
    com.apple.LaunchServices LSHandlers '({LSHandlerRoleAll =
    \"com.microsoft.outlook\"; LSHandlerURLScheme = mailto;})'" with
    administrator privileges

end copyToUserTemplateFolder

on copyToAnotherFolder()

    set theFolder to choose folder with prompt "Select a folder for your
        default Outlook for Mac files..."
    set theFolder to POSIX path of theFolder

    -- Get the path to the current home folder
    -- in the form of /Users/homefolder

    set homeFolder to POSIX path of (path to home folder)

    -- copy the Microsoft User Data folder from the current user's home
    folder
    -- to /System/Library/User Template/English.lproj/Documents

    set MUD to "\"" & homeFolder & "Documents/Microsoft User Data" &
        "\""
    set UTDDocuments to "\"" & theFolder & "/Documents/Microsoft User
        Data" & "\""
    set copyMUD to "cp -fR " & MUD & " " & UTDDocuments as string
    do shell script "mkdir -p " & "\"" & theFolder & "Documents" & "\""
    do shell script copyMUD

    -- copy the Microsoft preferences from the current user's home
    folder
    -- to /System/Library/User
    Template/English.lproj/Library/Preferences

    set MSPreferences to homeFolder &
        "Library/Preferences/*[Mm]icrosoft*"
    set UTPreferences to "\"" & theFolder & "Library/Preferences" & "\""
    set copyMSPreferences to "cp -fR " & MSPreferences & " " &
        UTPreferences as string
    do shell script "mkdir -p '" & theFolder & "Library/Preferences'"
    do shell script copyMSPreferences

    set emailDefaults to "defaults write '" & theFolder &
        "Library/Preferences/com.apple.LaunchServices' LSHandlers
        '({LSHandlerRoleAll = \"com.microsoft.outlook\";

```

```
        LSHandlerURLScheme = mailto;}}'" as string
    do shell script emailDefaults
end copyToAnotherFolder
```